
· Function modules are actively integrated modularization units with an interface.

· Function modules can, therefore, be called specifically from within programs or function modules.

· The interface of a function module can contain the following elements:

· Import parameters are passed to the function module. In general, these parameters are assigned standard ABAP Dictionary types. Import parameters can also be optional.

· Export parameters are received from the function modules. Export parameters are always optional.

· Changing parameters are passed to the function module and can be changed by it.

· Exceptions provide information about error situations. If a function module initiates an exception, processing of the function module is interrupted. It is possible to set a return code, if necessary, in the calling program. 

· As is the case with subroutines, a function module can contain local type and data object definitions.

· A function group represents the main program for function modules. Several function modules that operate on the same data content are combined to form a function group.

· The function group remains active for as long as the calling program is active. For example, if an executable program calls a function module, its entire function group is loaded as well.  It remains active until the executable program is completed.

· A function group can contain the same components as an executable program. These include:

· Data Objects
These are then global in relation to the function group, that is, they are visible to and changeable by all function modules within the group. The validity period is the same as for the function group.

· Subroutines
These can be called from all function modules in the group. 

· Screens
These can be called from all function modules in the group. 

· The global data of a function module is retained until the program that contained the first call of a function module in the function group is finished.

· Thus, if a function module that writes values to the global data is called, other function modules in the same function group can access this data when the program calls them.

· Seen from the outside, the global data is encapsulated, that is, it is not possible to access it directly. Therefore, you must have function modules that allow orderly access from the outside.

· The same applies to all the other components of the function group (screens, subroutines).

In the Attributes of a function module, you specify its general administrative data and the processing type:

· Remote-enabled function modules can be called asynchronously in the same system and can be called from other systems (not just R/3 Systems). To call a function module in another system, there must be a valid system connection. For further information, refer to the course BC415 - Communications Interfaces in ABAP or the online documentation.

· Update function modules contain additional functions for bundling database changes. For further information, refer to the course BC414 - Programming Database Updates or the online documentation.

The online documentation also details the interface restrictions that apply to remote-enabled and update function modules.

· Three kinds of parameters can be used to exchange data with function modules:

· Importing parameters, which are received by the function module 

· Exporting parameters, which are returned by the function module 

· Changing parameters, which are both received and returned

· By default, all parameters are passed by reference. You can change exporting and changing parameters only in the function module. This rule ensures that you avoid unwanted side effects. If you want to pass parameters by value, you must select the relevant option when you define the interface. 

· You can also declare importing and changing parameters as optional. You do not have to pass values to these parameters when you call the function module. Where possible, use this option when you add new parameters to function modules that are already in use. You can assign a default value to an optional parameter. If you do not pass a value of your own when you call the function module, the system then uses the default value instead. Exporting parameters are always optional.

· You may specify the type of an elementary parameter. You must specify the type of a structured or table parameter. You can use either ABAP Dictionary types, ABAP Dictionary objects, predefined ABAP type (i,f,p,n,c,string,x,xstring,d,t) or user-defined types. Any type conflicts show up in the extended program check. You can also assign a reference type.

· Table parameters are obsolete for normal function modules but have been retained to ensure compatibility for function modules with other execution modes. 

...

When you save the interface, the system generates the statement framework together with the comment block that lists the interface parameters: 
 
FUNCTION name.
*"--------------
*" ...
*"--------------

...

ENDFUNCTION.

The comment block is updated automatically if you make changes to the function module later on. This means that you can always see the interface definition when you are coding the function module. You program the statements exactly as you would in any other ABAP program in the ABAP Editor. 

In the function module, you can create your own local types and data objects and call subroutines or other function modules. 

You can make a function module raise exceptions.

To do this, you must first declare the exceptions in the interface definition, that is, assign each one a unique name. In the source code of your function module, you program the statements that raise an exception under the required condition. At runtime, the function module is terminated when an exception is triggered. The changes to exporting and changing parameters are the same as in subroutines. You can use two statements to raise an exception. In the forms given below, exception stands for the name of an exception that you declared in the interface. The system reacts differently according to whether or not the exception was listed in the function module call: 
· RAISE exception.
If the exception is listed in the calling program, the system returns control to it directly. If the exception is not listed, a runtime error occurs.

· MESSAGE <kind><num>(<id>) RAISING <exception>.
NOTE: If the exception is listed in the calling program, the statement has the same effect as RAISE <exception>. If it is not listed, the system sends message <num> from message class <id> with type <kind>, and no runtime error occurs.  

Function modules differ from subroutines in that you must assume that other programmers will use them. For this reason, you should ensure that you complete the steps listed here:
· Documentation (can be translated)
You should document all your parameters and exceptions, along with your entire function module, with short text (and long text if necessary). The system provides a text editor for you to do this, containing predefined sections for Functions, Example Call, Hints, and Further Information. 
· Work list
When you change an active function module, it acquires the status active (revised). When you save it, another version is created with the status inactive. When you are working on a function module, you can switch between the inactive version and the last version that you activated. When you activate the inactive version, the previous active version is overwritten.
· Function test
Note: Once you have activated your function module, you can test it using the built-in test environment in the Function Builder. If an exception is triggered, the test environment displays it, along with any message that you may have specified for it. You can also switch into the Debugger and the Runtime Analysis. You can save test data and compare sets of results.

[image: image12.wmf]ã

SAP

AG 2002

Call

DATA

:

result

TYPE s_

seatsmax

.

PARAMETERS:

pa

_

type

TYPE s_

planetye

,

pa

_

occ

TYPE s_

seatsmax

.

CALL FUNCTION 

'BC402

_

FMDD

_

GET

_

FREE

_

SEATS'

EXPORTING

ip_

planetype

=

pa

_

type

ip_

seatsocc

=

pa

_

occ

"

default

: 0

IMPORTING

ep

_

seatsfree

=

result

EXCEPTIONS

no

_

seats

= 1

overload

= 2

OTHERS

= 3.

CASE

sy

-

subrc.

WHEN

0.

WRITE: /

result COLOR

5.

WHEN

1.

WRITE: /

'You''ll have

to stand,

it''s

a

freighter

!'(

frt

).

WHEN

2.

WRITE: /

'The

plane

has already been overloaded

!'(

nos

).

WHEN

3.

WRITE: /

'Please contact your system administrator

!'(

adm

).

ENDCASE.

Insert pattern

CALL FUNCTION

. . .

. . .

. . .

. . .

. . .

BC402

_

FMDD

_

GET

_

FREE

_

SEATS

CALL FUNCTION

CALL FUNCTION

'

'

'

'

EXPORTING

EXPORTING

IMPORTING

IMPORTING

EXCEPTIONS

EXCEPTIONS

=

=

=

=

=

=

= 1

= 1

= 2

= 2

OTHERS

OTHERS

= 3.

= 3.

CASE

CASE

sy

sy

-

-

subrc.

subrc.

WHEN

WHEN

0.

0.

WHEN

WHEN

1.

1.

WHEN

WHEN

2.

2.

WHEN

WHEN

3.

3.

ENDCASE.

ENDCASE.


When you insert a function module call in your program, you should use the Pattern function. Then, you only need to enter the name of the function module (input help is available). The system then inserts the call and the exception handling (MESSAGE statement) into your program. 
You assign parameters by name. The formal parameters are always on the left side of the expressions:  


· Exporting parameters are passed by the program. If a parameter is optional, you do not need to pass it. Default values are displayed if they exist. 
· Importing parameters are received by the function module. Importing parameters are always optional.

· Changing parameters are both passed and received. You do not have to list optional parameters. Default values are displayed if they exist. 
The system assigns a value to each exception, beginning at one and continuing to number them sequentially in the order they are declared in the function module definition. You can assign a value to all other exceptions that you have not specifically listed using the special exception OTHERS. 
If you list the exceptions and one is triggered in the function module, the corresponding value is placed in the return code field sy-subrc. If you did not list the exception in the function call, a runtime error or a message occurs, depending on the statement you used in the function module to trigger the exception. 

As described in the ABAP Runtime Environment unit, the ABAP Workbench helps you to structure your source code when you work with function groups and function modules. 
Forward navigation ensures that you always enter the correct object. Include programs are named automatically, and the relevant call statements are inserted automatically in the correct positions.
You only have to observe the naming convention for function groups: Y|Z<rem_name>.

The system then creates a type F program called SAPLY|Z<rem_name>, which contains automatically generated INCLUDE statements. The include programs are also named automatically: LY|Z<rem_name><abbrev><num>.  

The system also inserts the include program, LY|Z<rem_name>UXX,  which contains an include statement in the form LY|Z<rem_name>U<num> for each function module.

· At first glance, a class is very much like a function group since it contains data objects (attributes) and functions (methods). These components can be protected from access by components of other programs. Then they can be accessed only by methods belonging to the class.

· In contrast to function groups, classes can have multiple instances, that is, more than one runtime object can be created for each program in a class. In other words, a class contains the technical description of objects (instances). All these objects, therefore, have the same attributes and provide the same methods. However, they can be distinguished physically. Their attributes can have different features. 

· Classes can be defined either locally within an executable program or globally in the Class Builder.

· The ABAP runtime system provides support for standard object-oriented syntax elements as of Release 4.6A. For more details, refer to the SAP online library.

[image: image16.wmf]ã

SAP AG 2001

ABAP

program

Creating Objects and Calling Methods

DATA ref1 TYPE REF TO

class

.

DATA ref2 TYPE REF TO

class

.

CREATE OBJECT ref1 ...

CREATE OBJECT ref2 ...

CALL METHOD ref1

-

>meth1

EXPORTING ...

IMPORTING ...

EXCEPTIONS ...

CALL METHOD ref2

-

>meth1

EXPORTING ...

IMPORTING ...

EXCEPTIONS ...

...

ref1

ref2

meth1

meth1


· To create instances of classes and to be able to access them, you need reference variables, which are pointers to these instances. You define these reference variables using DATA reference_name TYPE REF TO class_name. 

· At program start, the data objects are created in memory. However, since no instances have been created yet, the reference variables are still empty.

· At runtime, you can create as many instances of classes as you wish using
CREATE OBJECT reference_name.
If necessary, the import parameters of the special method CONSTRUCTOR must be filled with values. This special method is executed immediately after the instance is created. As a rule, basic settings for the actual instance are thus created.

· You can call methods of an instance using
CALL METHOD reference_name->method_name.
In contrast to a function module call, the method name is not sufficient here because, generally, several instances of a class exist for each program. Using the reference variable, followed by the object component selector ->, you instruct a particular instance to execute its method. 

· As of Release 4.6A, the ABAP Workbench offers several new EnjoySAP controls in order to move dialog functions from the application server to the presentation server. These controls are called on an object-oriented basis using classes, that is, the SAP Control Framework.

· For each method call, you select the controls from your ABAP program. The Control Framework sends your requests to the presentation server where they are converted on a platform-specific basis. 

· You can give the user the option of triggering events on the presentation server. The Control Framework then converts these events and you can react to them in your ABAP program. 

· For more details on the EnjoySAP controls and the Control Framework, refer to the SAP online library.

· A BAPI can be viewed as the door to the SAP system; this interface provides access to business data and processes in the SAP system from the outside. 

· Each object in the BOR can have several methods, one or more of which can be implemented as BAPIs.

· BAPIs usually exist for basic functions of a business object, such as:

· Creating objects

· Querying the attributes of an object

· Changing the attributes of an object

· A BAPI can be used for various applications. For example:

· Internet application components: Depiction of individual SAP R/3 functions in the Internet or intranet - for example, for users who do not have SAP R/3 experience.

· R/3 component creation: Communication between the business objects of different SAP R/3 components (solutions).

· Customers’ or partners’ code: External clients (for example, alternative GUIs) directly access business data and processes of the SAP R/3 System. In particular, BAPI calls (as calls of RPC-enabled function modules also) can be implemented in other programming languages. 

· There are standard methods for BAPIs with standardized names.

· Some of the most important standard method names are listed here.

· In Release 4.6C, BAPIs are implemented using function modules.

· From the SAP Easy Access menu, you call the BAPI Explorer by choosing Tools  Business Framework  BAPI Browser.

· Using the BAPI Explorer, you can navigate to the function module display for the selected BAPI in the Function Builder.

· Select the BAPI in the hierarchy section.

· In the detailed display window, choose the Detail tab.

· By double-clicking on the name of the function module, you can display it in the Function Builder.

· Function modules for BAPIs must fulfill the following requirements:

· Naming convention BAPI_<business_object_name>_<method_name> must be used
· Remote capability

· No user dialogs, neither screens nor messages, allowed

· Interface parameters must be typed with a Dictionary structure that was created for this BAPI

· Structures must comply with the naming convention BAPI_<structure_name> and are changed only on a compatible basis

· No CHANGING parameters allowed

· No exceptions allowed. Errors are reported to the user through the special export parameter RETURN.

· If you would like to use a BAPI in the SAP R/3 System, you can call the function module containing it.

· Pay attention to the limitations already mentioned.

· BAPI interfaces are created according to the needs for the "external" call, that is, from a non-R/3 System. Quantities are expected in an external format with 4 or 9 decimal places. The quantities must be transferred to the interface in converted format, even if the corresponding currency has no decimal places.

· For this conversion you can use function modules from the BACV function group (development class SBF_BAPI).

	At the conclusion of these exercises, you will be able to:

· Create and implement function groups

· Implement function modules

· Call function modules

	You are a developer for an airline consortium. Your task is to develop evaluation programs for several airlines.


1.
All plane types that are available for each airline should be stored in a function group in an internal table. For simplicity, these should have a flat line structure and not a nested structure.
## is your two-digit group number
Model solution:
TAW10_BASICS_FLIGHT


1-1
Create function group Z##_TAW10_FLIGHT.

1-2
Assign it to message class TAW10.

	You can find the program ID in the "TOP include" (LTAW10_BASICS_FLIGHTTOP).


1-3
Document your function group.

2.
A function module should fill internal tables for plane types. Only the available replacement types for the airline with sufficient seats should be written to the table.
From the total prices that are transferred, the average price for each seat should also be calculated. The plane list should then be sorted in descending order before it is returned.
## is your two-digit group number
Model solutions:
TAW10_BASICS_FLIGHT
TAW10_BASICS_CREATE_PLANELIST


2-1
Create function module Z_##_TAW10_CREATE_PLANELIST within your function group Z##_TAW10_FLIGHT.

2-2
Declare the line type t_carr_plane as your function group’s global data type. It should be structured as follows:


	Component
	Type

	carrid
	scarplan-carrid

	planetype
	scarplan-planetype

	seatsmax
	saplane-seatsmax



The line type assigns the airline to the plane type.

2-3
Declare the internal table it_carr_planes with line type t_carr_plane as the global data object of your function group. It should be in the form of a sorted table with the unique key carrid and planetype.

2-4
Fill the internal table it_carr_planes for each array fetch using view TAW10_CARPLAN.

	Choose an appropriate time for the event to occur. Remember that a function group cannot be executed directly.
Implement the corresponding event block in a suitable include program that you have added to the main program at a suitable location. Adhere to the naming conventions for function group include programs.


2-5
Now declare the import parameter with pass by value for your function module:
ip_seatsocc as an optional parameter with the default value 0,
ip_carrid, ip_paymentsum, ip_currency.

2-6
Declare the export parameter ep_planelist with pass by value. Assign a type using the global table type TAW10_TYPS_PLANETAB.

2-7
Declare the exception no_planes and document it.

2-8
In the function module, create local structure l_wa_carr_plane of type t_carr_plane.

2-9
From the global internal table, read those plane types that are available to the airline and which have sufficient seats for the booking that was transferred.
In this loop calculate the average price for each seat for each plane type. Also declare work area l_wa_plane as a local data object within the function module. Assign a type using the global structure TAW10_TYPS_PLANE.
Attach the completely filled work area to the internal table to be exported.

2-10
Before exporting, sort the internal table by the average price per seat.

2-11
Trigger the exception if no suitable plane types could be found. Implement the trigger using error message 040, specifying the airline.

2-12
Document your function module.

2-13
Test your function module.

3.
If a plane breaks down that is already scheduled for a flight and already has bookings, a replacement plane must be used.  Write a program that outputs all the available replacement plane types for each airline in a list. You can obtain the data for the list using the function module you created in Task 2.
Model solution:
TAW10_BASICS_CALL_FUNCTION


3-1
Create a new program Z##_TAW10_SUBSTITUTE_PLANES.

3-2
In the program, define an internal table it_planelist that stores the replacement plane types. The type of this internal table results from the function module interface.

3-3
The user should query the IMPORTING parameters of the function module using a selection screen, where the field for the airline must be a mandatory field.

3-4
Call your function module Z_##_TAW10_CREATE_PLANELIST (use Sample) and populate the interface. Catch the exceptions with appropriate error messages (error messages 40 and 41 of message class TAW10). Ensure that users return to the selection screen when an exception occurs.

3-5
Display the table it_planelist filled by the function module in a list. Use a field symbol for this.


Function Groups and Function Modules Solutions

1-1, 2-4


Model solution SAPLTAW10_BASICS_FLIGHT

**************************************************************
*   System-defined Include-files.                                 *
**************************************************************
  INCLUDE ltaw10_basics_flighttop.       " Global Data
  INCLUDE ltaw10_basics_flightuxx.       " Function Modules

**************************************************************
*   User-defined Include-files (if necessary).                    *
**************************************************************
  INCLUDE ltaw10_basics_flighttop.             " Subprograms
* INCLUDE LTAW10_BASICS_FLIGHTO...             " PBO-Modules
* INCLUDE LTAW10_BASICS_FLIGHTI...             " PAI-Modules

  INCLUDE ltaw10_basics_flighte01.       " Events




1-2, 2-2, 2-3

Model solution LTAW10_BASICS_FLIGHTTOP

FUNCTION-POOL TAW10_BASICS_FLIGHT MESSAGE-ID taw10.


TYPES:
  BEGIN OF t_carr_plane,
    carrid    TYPE scarplan-carrid,
    planetype TYPE scarplan-planetype,
    seatsmax  TYPE saplane-seatsmax,
  END OF t_carr_plane.


DATA:
  it_carr_planes TYPE SORTED TABLE OF t_carr_plane
                 WITH UNIQUE KEY carrid planetype.


2-4




Model solution LTAW10_BASICS_FLIGHTE01

*------------------------------------------------------------*
*   INCLUDE LTAW10_BASICS_FLIGHTE01                          *
*------------------------------------------------------------*

  LOAD-OF-PROGRAM.

    SELECT carrid planetype seatsmax
           FROM taw10_carplan
           INTO CORRESPONDING FIELDS OF TABLE it_carr_planes.


2-1, 2-5 – 2-11

Model solution TAW10_BASICS_CREATE_PLANELIST

FUNCTION TAW10_BASICS_CREATE_PLANELIST.
*"------------------------------------------------------------
*"*"Local interface:
*"  IMPORTING
*"     VALUE(IP_SEATSOCC) TYPE  SFLIGHT-SEATSOCC DEFAULT 0
*"     VALUE(IP_CARRID) TYPE  SPFLI-CARRID
*"     VALUE(IP_PAYMENTSUM) TYPE  SFLIGHT-PAYMENTSUM
*"     VALUE(IP_CURRENCY) TYPE  SFLIGHT-CURRENCY
*"  EXPORTING
*"     VALUE(EP_PLANELIST) TYPE  TAW10_TYPS_PLANETAB
*"  EXCEPTIONS
*"      NO_PLANES
*"------------------------------------------------------------

  DATA:
    l_wa_carr_plane TYPE t_carr_plane,
    l_wa_plane      TYPE taw10_typs_plane.

  LOOP AT it_carr_planes INTO l_wa_carr_plane
                         WHERE carrid EQ ip_carrid
                         AND   seatsmax GE ip_seatsocc.
    l_wa_plane-planetype = l_wa_carr_plane-planetype.
    l_wa_plane-seatsmax  = l_wa_carr_plane-seatsmax.
    l_wa_plane-avg_price =
           ip_paymentsum / l_wa_carr_plane-seatsmax.
    l_wa_plane-currency  = ip_currency.
    APPEND l_wa_plane TO ep_planelist.
  ENDLOOP.

  IF sy-subrc NE 0.
    MESSAGE e040 RAISING no_planes WITH ip_carrid.
  ELSE.
    SORT ep_planelist BY avg_price DESCENDING.
  ENDIF.

ENDFUNCTION.


3
Model solution TAW10_BASICS_CALL_FUNCTION

*&-----------------------------------------------------------*
*& Report  TAW10_BASICS_CALL_FUNCTION                        *
*&                                                           *
*&-----------------------------------------------------------*
*& solution of exercise 3 function groups                    *
*&                        and function modules               *
*&-----------------------------------------------------------*

REPORT  taw10_basics_call_function.

DATA:          it_planelist TYPE taw10_typs_planetab.

FIELD-SYMBOLS:  <plane> TYPE LINE OF taw10_typs_planetab.

PARAMETERS: pa_carr  TYPE sflight-carrid OBLIGATORY,

            pa_occ   TYPE sflight-seatsocc,

            pa_paysu TYPE sflight-paymentsum,

            pa_curr  TYPE sflight-currency DEFAULT 'EUR'.

AT SELECTION-SCREEN.

  CALL FUNCTION 'TAW10_BASICS_CREATE_PLANELIST'

    EXPORTING

      ip_seatsocc   = pa_occ

      ip_carrid     = pa_carr

      ip_paymentsum = pa_paysu

      ip_currency   = pa_curr

    IMPORTING

      ep_planelist  = it_planelist

    EXCEPTIONS

      no_planes     = 1

      OTHERS        = 2.

  CASE sy-subrc.

    WHEN 1.

      MESSAGE e040(taw10) WITH pa_carr.

    WHEN 2.

      MESSAGE e041(taw10).

  ENDCASE.

START-OF-SELECTION.

  LOOP AT it_planelist ASSIGNING <plane>.

    WRITE: / <plane>-planetype,

             <plane>-seatsmax,

             <plane>-avg_price CURRENCY <plane>-currency,

             <plane>-seatsmax,

  ENDLOOP.

